
Objectives

• Determine how female choice affects allele and genotype
frequencies in a population.

• Determine how initial allele frequencies influence the evolu-
tion of allele frequencies through female choice.

• Evaluate how natural selection can counter sexual selection
in the evolution of a trait.

Suggested Preliminary Exercises: Hardy-Weinberg Equilibrium
and Multilocus Hardy Weinberg

INTRODUCTION
From a genetic perspective, evolution is often described as a change in allele fre-
quency over time. What mechanisms cause changes in allele frequencies? Gene
flow, mutations, and genetic drift can all spur such change. Natural selection—
the differential survival and reproductive success of individuals in populations—
is another major evolutionary force. Natural selection simply means that if some
individuals have genetic characteristics that are well-suited for a particular envi-
ronment, they will on average survive better and produce more offspring than
other individuals in the population, thereby changing allele frequencies in sub-
sequent generations.

In some cases, natural selection arises from differences in mating success: cer-
tain individuals possess traits that cause them to be perceived as “better” mates,
and hence to mate more frequently than other individuals in the population.
For example, the long, bright tails of male peacocks may have evolved because
females preferentially selected males with the longest and brightest tails (the selec-
tive force was female choice). This difference in mating success due to such traits
is called sexual selection.

Charles Darwin  thought that sexual selection was different from natural selec-
tion, saying “Sexual selection … depends not on a struggle for existence, but on a
struggle between the males for possession of the females; the result is not death to
the unsuccessful competitor, but few or no offspring” (Darwin 1871).

The theory of sexual selection assumes the selection of traits that are purely con-
cerned with maximizing mating success. Males can “increase the odds” of mating
by having traits (such as the long, bright tail feathers in male peacocks) that attract
females. Males can also maximize their mating success by the “brute force” method:



outcompeting other males for mating opportunities (male-male contests). Female traits
may not be so visible; females maximize their fitness by selecting males that somehow
enhance their own fitness or the fitness of their offspring. A female might select males
that have “good genes” which enhance her offspring’s fitness (an indirect benefit of mate
choice), or by selecting males that are “good parents/mates,” which enhance the female’s
own survival and reproductive success (a direct benefit of mate choice).

In these cases it’s fairly easy to imagine how females that choose beneficial mates can
be favored in a population, and how such choices influence the evolution of a species
(Alcock 2001). But what happens when there is no direct or indirect fitness benefit asso-
ciated with mate choice? Can a population still evolve due to sexual selection? The
answer, in theory, is yes. Ronald Fisher introduced the theoretical argument in 1930.
Fisher realized that sexual selection could cause populations to evolve when there is no
fitness gain associated with mate choice, and that sometimes even traits that decrease
survivorship, such as an extraordinarily long tail, can evolve in a population as a result.
Fisher’s model is called runaway sexual selection.

An important underlying assumption of Fisher’s model is that both the female pref-
erence and the male trait (i.e. tail length) must be under genetic control. (Remember,
traits cannot evolve unless they have a genetic basis.) So, let’s imagine that males have
a gene associated with tail length in which males have either a T1 (short) or a T2 (long)
genotype (let’s assume, for simplicity, that we’re dealing with a haploid organism).
Females also have these genes for tail length but do not express them. Let’s further
assume that the T2 genotype has a fitness cost—perhaps males with long tails have higher
mortality rates because predators capture them more easily. Let’s also imagine that a
separate, nonlinked gene determines mating preference, where the genotype P1 indicates
no preference for tail length but the P2 allele indicates a strong preference for long tails.
Both males and females have the P gene, but only females express the gene when they

Figure 1 When some females prefer males with
long tails, males with the T2 genotypes will
increase in frequency in the population in the
next generation (bottom). P1 females randomly
choose to mate with both long- and short-tailed
males, while P2 females prefer males with long
tails. If this preference is strong enough, and if P2
females are sufficiently frequent in the popula-
tion, long-tailed males may mate more success-
fully on average and thus produce more off-
spring than short-tailed males. These offspring
will tend to inherit both the allele for long tails
(from their male parents) and the allele for tail
preference (from their female parents), so that as
selection increases the frequency of T2 it also
increases the frequency of P2. As P2 becomes
more frequent and an increasing proportion of
females favors long tails, the advantage of hav-
ing a long tail increases. Alleles T2 and P2 may
thus both increase in frequency over time, at
ever-increasing rates. The change in genotype
frequency over time for males is shown in Figure
2. Note that male genotypes containing a T2 allele
increase in frequency, while male genotypes with
T1 decrease in frequency. (From Futuyma 1998.)



solicit matings. Thus, both sexes carry an allele for both the P and T genes. Because of
this, selection for a particular allele of one gene can “drag along” a particular allele of
the other gene. This association leads to a genetic correlation between the tail length
gene and the mating preference gene, as shown in Figure 1.

If runaway sexual selection actually happens in nature, why don’t we commonly see
birds with super-long tails? Although sexual selection for long tails increases the fitness
of long-tailed males over short-tailed males, natural selection may select against long-
tailed males through decreased survivorship. For example, if a tail is so long that the
bird has troubles escaping from predators, there will be fewer long-tailed males in the
population. Depending on the strength of selection against long-tailed males, we can
expect some equilibrium level that would balance survival costs of having a long tail
with the reproductive benefits of having such a tail. Figure 3 shows an example of how
natural selection can drive the T2 allele to extinction by substantially decreasing the sur-
vival probability of T2 males.

INSTRUCTIONS

A. Set up the model
parental population.

1. Open a new spread-
sheet and set up headings
as shown in Figure 4.

2. In cells C5–C8 and
E5–E8, enter the number
of individuals with each
genotype. 

In Fisher’s (1958) model of sexual selection, he assumed that the female’s prefer-
ence must confer some sort of selective advantage and that this advantage was neces-
sary to “get the ball rolling” in the runaway process. However, later work by Kirkpatrick
(1982) showed that an initial selective advantage was not a necessary prerequisite for
the runaway process of sexual selection and that evolution of the male trait could occur
without selection for or against female preference. We will model Fisher’s runaway
process of sexual selection and in doing so, show that no initial selective advantage of
female preference is necessary for generating the runaway process.

PROCEDURES

In this exercise, you’ll set up a runaway sexual selection model and see first-hand how
the runaway process works. You’ll model a population of 2000 haploid individuals
(1000 males and 1000 females). There will be two alleles, P and T, as previously
described, and thus there are 4 possible genotypes: P1T1, P1T2, P2T1, and P2T2. You
will set up a table of mating preferences that indicate the preferences of a female geno-
type for the various male genotypes. These mating preferences will be converted to
mate selection probabilities that account for the frequencies of male genotypes in the
population. Once the mating preferences are assigned, you will simulate the repro-
duction of offspring as a diploid organism (by joining the male and female partner’s
genotypes), and then will simulate meiosis to so that organisms so that the haploid sys-
tem is maintained. Once the offspring are generated, you will compute the numbers of
P1T1, P1T2, P2T1, and P2T2 individuals in the offspring population, and then compute
their allele frequencies. And finally, you will develop a macro to track the allele fre-
quencies over time to see how the various genotypes evolve.

Admittedly, this is a pretty complicated spreadsheet, so take your time as you work
through it and try to keep the bigger picture in mind as you develop the model. As
always, save your work frequently to disk.

ANNOTATION

We will start by setting up a parent population that contains 1000 males and 1000
females. The tail length locus, T, has 2 alleles, T1 (short tail) and T2 (long tail). The pref-
erence locus, P, has two alleles, P1 (no preference for tail length) and P2 (preference
for long tails).

The possible genotypes for the population are given in cells A5–A8.

Enter 250 in cells C5–C8 and E5–E8.
To begin, we will assume that the all genotypes are equally represented in the popu-
lation. You will be able to change these cells later in the exercise. 

1
2
3
4
5

7
8

A B C D E F
Sexual Selection

Tally Tally

Genotype Male 0 Female 0

T1P1 Short tail, no preference 250 250

T1P2 Short tail, preference 250 250

T2P1 Long tail, no preference 250 250

T2P2 Long tail, preference 250 250

PARENTAL GENOTYPE NUMBERS

Figure 4



This is a “place holder” to tally the total number of males and females in cells D5–D8
and cells F5–F8. It is necessary so that we can assign genotypes properly to the 1000
males and females. 

Enter =C5 in cell D5.
Enter the formula =D5+C6 in cell D6. Copy this formula down to cell D8. 
This is a running tally that counts the total number of individuals as we consider addi-
tional genotypes. The final result in cell D8 should be 1000. 

Enter =E5 in cell F5. 
Enter =F5+E6 in cell F6. Copy the formula down to cell F8. Your total should be 1000
in cell F8.

Enter 0 in cell A22. 
In cell A23, enter =1+A22. Copy your formula down to cell A1021. This assigns a num-
ber to each male and each female in the population.

Enter the formula =LOOKUP(A22,$D$4:$D$8,$A$5:$A$8). Copy this formula down
to cell B1021.
The LOOKUP function looks up a value (A22) in a vector that you specify (cells
$D$4:$D$8), and returns a genotype for the individual given in the vector $A$5:$A$8.
(A vector is a single row or column of values). The result of this function is that geno-
types are assigned to individuals in exactly the numbers that you specified in cells
C5–C8.

Examine your first 10 genotypes. They should all be T1P1. To see how the function
works, change cell C5 to 1. Now examine the genotypes of your first 10 individuals.
The first male should be T1P1, but the rest of the males should be T1P2. When you feel
you have a handle on how this function works, return cell C5 to 250. 

Enter the formula =LOOKUP(A22,$F$4:$F$8,$A$5:$A$8) in cell C22. Copy your for-
mula down to cell C1021.
The formula for females works in the same way as that for males, using the female
tallies.

3. Enter 0 in cells D4 and
F4. 

4. In cell D5–D8, enter for-
mula to tally the male
genotypes.

5. Set up the tally for
females in cells F5–F8.

6. Set up new spreadsheet
headings as shown in
Figure 5. 

7. Set up a linear series
from 0 to 999 in cells
A22–A1021.

8. In cells B22–B1021, use
the LOOKUP function to
assign genotypes to the
males.

9. In cells C22–C1021,
enter a formula to assign
genotypes to the females.

10. Save your work.

B. Set up the mating
preferences and mate
selection probabilities. 

1. Set up new column
headings as shown in
Figure 6.

2. Enter the female mate
selection preferences
shown in cells I5–L8. 

Cells H5–H8 represent the genotypes of females, and cells I4–L4 represent the geno-
type of a female’s potential mate. The entries in cells I5–L8 establish the female’s mat-
ing preferences. Thus, a female with genotype T1P1 has the “no preference for tail length
gene,” so her preferences are identical for all four male genotypes. A female with geno-
type T1P2 or T2P2 has the P2 “preference for long tailed males gene,” so she will prefer
to mate with males that have a genotype T2P1 or T2P2, but will not prefer males with
genotypes T1P1 or T1P2. Note that the probabilities in each row in this table must sum
to 1! 

Enter the formula =I5*C5/(I5*C5+J5*C6+K5*C7+L5*C8) in cell I14.
Although female mating preferences have been established, mating probabilities must
also consider the number of males of each genotype in the population. The formula in
cell I14 makes this adjustment and computes the probability that a T1P1 female will
mate with a T1P1 male. The formula multiplies the preference for T1P1 males by the
number of T1P1 males in the population, then adjusts this result by dividing by prefer-
ence × number for all of the genotypes in the population.

Double-check your formulae against Figure 7.

3. In cell I14, enter a for-
mula to compute the prob-
ability that a mating
between the specified
genotypes will take place.

4. Enter formulae to com-
pute the remaining mate
selection probabilities. 

G H I J K L

T1P1 T1P2 T2P1 T2P2

T1P1 0.25 0.25 0.25 0.25

Female T1P2 0 0 0.5 0.5

genotype T2P1 0.25 0.25 0.25 0.25

T2P2 0 0 0.5 0.5

T1P1 T1P2 T2P1 T2P2

Survival => 1 1 1 1

T1P1

Female T1P2

genotype T2P1

T2P2

Male genotype

MATE SELECTION PREFERENCES

MATE SELECTION PROBABILITIES

Male genotype

Figure 6

K L
=K5*C7/(I5*C5+J5*C6+K5*C7+L5*C8) =L5*C8/(I5*C5+J5*C6+K5*C7+L5*C8)

=K6*C7/(I6*C5+J6*C6+K6*C7+L6*C8) =L6*C8/(I6*C5+J6*C6+K6*C7+L6*C8)

=K7*C7/(I7*C5+J7*C6+K7*C7+L7*C8) =L7*C8/(I7*C5+J7*C6+K7*C7+L7*C8)

=K8*C7/(I8*C5+J8*C6+K8*C7+L8*C8) =L8*C8/(I8*C5+J8*C6+K8*C7+L8*C8)

14
15
16
17

I J
=I5*C5/(I5*C5+J5*C6+K5*C7+L5*C8) =J5*C6/(I5*C5+J5*C6+K5*C7+L5*C8)

=I6*C5/(I6*C5+J6*C6+K6*C7+L6*C8) =J6*C6/(I6*C5+J6*C6+K6*C7+L6*C8)

=I7*C5/(I7*C5+J7*C6+K7*C7+L7*C8) =J7*C6/(I7*C5+J7*C6+K7*C7+L7*C8)

=I8*C5/(I8*C5+J8*C6+K8*C7+L8*C8) =J8*C6/(I8*C5+J8*C6+K8*C7+L8*C8)

Figure 7



Double-check your results as well. Since there are currently equal numbers of male
genotypes in the population, the mate selection probabilities should be the same as the
mate selection preferences (Figure 8).

Enter the number 1 in cells I13–L13.
Currently the survival probability is set to 1 so that all male genotypes have equal
survival. Later in the exercise, you will be able to change these values so that males
with long tails have a lower probability of survival.

Our goal is to have the spreadsheet look up the genotype of female parents (in column
C) and match their genotype to genotypes listed in cells H14–H17. Ultimately, we want
to determine the genotype of the female’s selected mate, listed in cells I12–L12. To
choose mates according to the probabilities given, we will use four different functions:
MATCH, INDEX, RAND, and IF. The combination of these formulae will allow us to
generate the genotype of a mate for each female in the population in column J. 

The MATCH formula returns the relative position of an item in a table that matches the
condition you specify. The MATCH function has the syntax MATCH (lookup_
value,lookup_array,match_type), where lookup_value is the value you use to find
the value you want in a table,  lookup_array is a contiguous range of cells containing
possible lookup values, and match_type tells the spreadsheet how to match the lookup
value to the lookup array (by not specifying match-type, the default is used). In cell D22,
the formula =MATCH(C22,$H$14:$H$17) tells the spreadsheet to find the genotype
listed in cell C22, and return the relative position of that genotype in the $H14–$H17 table.
For example, the genotype of female 1 in the spreadsheet is T1P1. Excel returns the value
1 to indicate that T1P1 individuals occupy the first position in our table. If female 1 had
the genotype T2P2, the program would return the number 4 (fourth position).

The INDEX formula returns the value of an element in a table, once you identify the
row and column number that should be returned. The INDEX formula has the syntax:
INDEX(array,row_num,column_num), where array is a range of cells in a table;
row_num selects the row in the table from which to return a value, and column_num

5. Enter a survival proba-
bility for males in cells
I13–L13. 

6. Save your work.

C. Simulate parental
matings. 

1. Set up new headings as
shown in Figure 9.

2. In cell D22, enter the for-
mula =MATCH(C22,$H$14:
$H$17).

3. In cell E22, enter the for-
mula =INDEX($H$14:
$L$17,D22,2).

selects the column in table from which to return a value. In cell E22, the formula
=INDEX($H$14:$L$17,D22,2) tells the spreadsheet to examine the range of cells in
H14–L17 and go to the row designated in cell D22 (derived from your MATCH for-
mula) and column 2 (which indicates the probability of mating with an T1P1 male). The
spreadsheet will then return the value associated with this row and column intersec-
tion. Your result should be 0.25. 

Enter the formula =INDEX($H$14:$L$17,D22,3) in cell F22.
Enter the formula =INDEX($H$14:$L$17,D22,4) in cell G22.
Enter the formula =INDEX($H$14:$L$17,D22,5) in cell  H22.
These four formulae simply generate the appropriate mating probabilities for each indi-
vidual in the population.

Enter =RAND() in cell I22.
This formula randomly determines the genotype of the mate for each individual in the
population. When you press F9, the calculate key, you generate a new set of random
numbers.

Enter the formula =IF(I22<=E22,$I$12,(IF(I22<=E22+F22,$J$12,
(IF(I22<=E22+F22+G22,$K$12,$L$12))))) in cell J22.
The formula in cell J22 looks complicated but really it’s not. The formula is simply four
nested IF statements. The formula tells the spreadsheet to examine cell I22 (the ran-
dom number). If I22 is less than or equal to the value in cell E22 (<=E22), then return
the genotype in cell $I$12; otherwise walk through the next IF statement. The next state-
ment examines cell I22, and if its value is less than or equal to the values in cells E22 +
F22 (<=E22+F22),  then return the genotype in cell $J$12; otherwise walk through the
third  IF statement. The third statement examines cell I22, and if its value is less than
or equal to the sum of E22, F22, and G22 (<=E22+F22+G22), return the genotype in cell
$K$12, otherwise return the value in cell $L$12.

This will establish the selected mate’s genotype for each female in the population.
Review your spreadsheet entries and results for the first five individuals and make sure
you understand how mates were determined for the females.

We set up the spreadsheet so that selection against a particular genotype occurs after
female mating probabilities have been established. Thus, selection against a genotype
does not influence the mating probabilities themselves. For now, each genotype has a
survival probability of 1 (given in cells I13–L13), indicating that there is no “cost” to
having a long tail. If we wished to impose selection against long-tailed males, we would
alter the survival probabilities in cells I13–L13.

Enter the formula =IF(RAND()<HLOOKUP(J22,$I$12:$L$13,2),J22,”.”) in cell K22.
Copy the formula down to cell K1021.
The formula uses an HLOOKUP function to find the genotype of the selected mate for
female 1 (J22) in the table of cells I12–L13, and finds that male’s survival probability
in the second row of the table. The RAND() function draws a random number between
0 and 1. The IF function determines whether this random number is less than the appro-
priate survival probability. If the random number is less than the survival probability,

4. Use the INDEX function
to index the T1P2, T2P1,
and T2P2 genotypes in
cells F22–H22.

5. In cell I22, use the
RAND function to gener-
ate a random number
between 0 and 1.

6. In cell J22, enter a for-
mula to establish the geno-
type of that female’s
selected mate.

7. Select cells D22–J22, and
copy and the formula
down to row 1021

8. Save your work.

D. Impose natural selec-
tion on males. 

1. Set up new headings as
shown in Figure 10.

2. In cells K22–K1021 enter
a formula to compute
which males survive to
breed and produce off-
spring.

the male lives and his genotype (J22) is returned. If the random number is greater than
the survival probability, the male dies and a period (“.”) is returned, indicating a death.

Enter the formula =IF(K22=“.”,”.”,C22&J22) in cell L22.
If the male in cell K22 is dead, the formula returns a missing value (.). If the male is not
dead, the spreadsheet returns the combination of cells C22 and J22; the & function sim-
ply concatenates the two cells. 

Enter the formula =IF(RAND()<0.5,MID(L22,1,2),MID(L22,5,2))
&IF(RAND()<0.5,MID(L22,3,2),MID(L22,7,2)) in cells M22 and N22.
Our goal is to generate male and female offspring that have a single T allele and a sin-
gle P allele. We’ll let meiosis occur with random segregation of alleles.

The MID function returns a specific number of characters from a text string, starting
at the position you specify, and based on the number of characters you specify. It has
the syntax MID(text,start_num,num_chars) The first part of the formula,
IF(RAND()<0.5,MID(L22,1,2),MID(L22,5,2), tells the spreadsheet to draw a random
number between 0 and 1. If that random number is <0.5, return the value associated
with MID(L22,1,2), otherwise returns the value associated with MID(L22,5,2). The
MID(L22,1,2) portion of the formula tells the spreadsheet to examine cell L22 and, start-
ing at the first character, return 2 characters. The MID(L22,5,2) portion of the formula
examines cell L22, and starting at the fifth character, returns 2 characters. This portion
of the formula returns a randomly selected T allele. The second IF statement is analo-
gous and randomly selects the P allele for each offspring. The two alleles are joined
by the & symbol.

3. Save your work.

E. Establish offspring
genotypes and allele 
frequencies.

1. Set up new column
headings as shown  in
Figure 11.

2. In cell L22, enter a formu-
la to combine the female’s
haploid genotype with her
mate’s haploid genotype to
produce a diploid offspring
(only if the male survived
to breed).

3. In cells M22 and N22,
use the IF, RAND, and
MID functions to generate
the genotypes of haploid
inviduals. 

4. Select cells L22–N22 and
copy their formulae down
to row 1021.

5. Set up new column
headings as shown in
Figure 12.

Sexual Selection 491

19
20
21

L M N

Diploid Male haploid Female haploid 

genotype genotype genotype

Offspring population

Figure 11

9
10
11
12
13
14
15
16

A B C D E F

Male Female Male Female

T1P1 Short tail, no preference

T1P2 Short tail, preference

T2P1 Long tail, no preference

T2P2 Long tail, preference

OFFSPRING

Genotype numbers Genotype frequencies

Figure 12



Double-check your formulae against Figure 13.

Double-check your formulae against Figure 14.

The macro needs to perform the following steps:
•Paste the genotype numbers of the offspring into the parental population cells.
•Press the calculate key to simulate mate selection, natural selection, and breed-

ing.
•Record the offspring’s allele frequencies in the cells O4–V18.

There are many ways to write a macro to conduct these steps. We suggest one way, but
you may see other (perhaps easier) steps. Put your macro function in the “record macro”
mode and assign a shortcut key (see Exercise 2). Record the following operations:

6. In cells C12–C15 and
D12–D15, use the COUN-
TIF function to count the
number of offspring male
and female genotypes,
respectively. Sum the
totals in cells C16 and
D16.

7. In cells E12–F16, com-
pute the male and female
offspring genotype fre-
quencies.

8. Save your work.

F. Track genotype fre-
quencies over time.

1. Set up new headings as
shown in Figure 15, but
extend your generations to
15. 

2. Open Tools | Options |
Calculation and set the cal-
culation key to manual.

3. Write a macro to track
allele frequencies over
time.

=COUNTIF($M$22:$M$1021,A12) =COUNTIF($N$22:$N$1021,A12)

=COUNTIF($M$22:$M$1021,A13) =COUNTIF($N$22:$N$1021,A13)

=COUNTIF($M$22:$M$1021,A14) =COUNTIF($N$22:$N$1021,A14)

=COUNTIF($M$22:$M$1021,A15) =COUNTIF($N$22:$N$1021,A15)

=SUM(C12:C15) =SUM(D12:D15)

Genotype numbers

=C12/$C$16 =D12/$D$16

=C13/$C$16 =D13/$D$16

=C14/$C$16 =D14/$D$16

=C15/$C$16 =D15/$D$16

=SUM(E12:E15) =SUM(F12:F15)

Genotype frequencies

Figure 14

•Press F9, the calculate key, to generate new random numbers (and hence new
matings and offspring for the parental population).

•Select cells E12–E15. Copy.
•Select cell O3.
•Open Edit | Find. The dialog box in Figure 16 will appear. Leave the Find What

box blank and Search By Columns. Select Find Next and then Close. Your cursor
should move down to the next blank cell. 

•Open Edit | Paste Special and select the Paste Values and Transpose options. Click
OK.

•Select cells F12–F15. Copy.
•Select cell S3.
•Open Edit | Find.
•Click on Find Next and then Close.
•Open Edit | Paste Special,and select the Paste Values and Transpose options.
•Select cells C12–C15. Copy.
•Select cell C5. 
•Open Edit | Paste Special and select the Paste Values option.
•Select cells D12–D15. Copy.
•Select cell E5.
•Open Edit | Paste Special and select the Paste Values option.

Stop recording. Now when you press your shortcut key, your macro will record the
allele frequencies of the various genotypes for males and females.

Use the line graph option and label your axes fully. Your graphs should resemble Fig-
ures 17 and 18.

4. Run the macro 15 times
(i.e., over 15 generations).

5. Save your work.

G. Create graphs.

1. Graph the allele fre-
quencies of males and
females over time. Make a
separate graph for each
sex.

Figure 16



QUESTIONS

1. Interpret your model results. For each sex, explain how the genotypes (T1P1,
T1P2, T2P1, T2P2) evolve (change in frequency) from one generation to the next.
Which genotypes went extinct; which genotypes persisted? Did this differ for
males and females? If so, why? What mechanism allows for the evolution of the
T and P alleles? 

2. In your model, females with the P2 allele mate only with the T2 males—no
exceptions. In reality, perhaps not all females will be able to mate with T2 males,
and so some P2 females will mate with T1 males. Change the choice parameters

2. Save your work. 

Change in Genotype Frequencies: Males

Change in Genotype Frequencies: Females

in cells I6–L6 and I8–L8 to 0.1, 0.1, 0.4, and 0.4. Reset your genotype numbers in
cells C5–C8 and E5–E8 to 250. Clear your old macro results, and run your
macro again. How does the “strength” of sexual selection influence the change
in allele frequencies from one generation to the next? 

3. How does natural selection influence the evolution of the T2 trait even when P2
females have full preference for T2 male? Return the mate selection preferences
to their original values as shown:

Decrease the survival probabilities of T2 males in cells K13–L13 by increments
of 0.1. With each incremental decrease in survival, run your macro again (clear
your old results, and make sure to reset the initial genotype numbers to 250 in
cells C5–C8 and E5–E8). What level of natural selection “puts the brakes” on
sexual selection? 

4. How do starting allele frequencies affect the outcome of a simulation? The ini-
tial genotypes we used to build the spreadsheet are admittedly very unusual;
before sexual selection for tail length begins, it is much more likely that at least
one (if not both) of the alleles T2 and P2 will be new and very rare mutations.
That is, either there will be variety in tail length (long and short tails both occur
with some regularity) when a mutation causes one female (or a few sisters) to
prefer long tails, or there will already be a preference for a trait that does not yet
exist, and mutation will create that trait in one male (or a few brothers). We can
use this spreadsheet model to test both of these initial conditions.

Set the genotype survivals back to 1 and set the initial genotype numbers as
shown below: 

In this population, half the males have long tails and half have short, but none
of the males carry the allele for preferring long tails. Approximately half the
females carry the allele for long tails, but by some unusual chance, 10 sisters in
this generation all received a mutated gene that causes them to mate exclusive-
ly with long-tailed males. Clear your previous results from cells O4–V18 and
run your macro to see what happens to genotype frequencies over 15 genera-
tions. Do these initial conditions result in runaway sexual selection?

2
3
4
5
6
7
8

G H I J K L

T1P1 T1P2 T2P1 T2P2

T1P1 0.25 0.25 0.25 0.25

Female T1P2 0 0 0.5 0.5

genotype T2P1 0.25 0.25 0.25 0.25

T2P2 0 0 0.5 0.5

Male genotype

MATE SELECTION PREFERENCES

3
4
5
6
7
8

A B C D E F
Tally Tally

Genotype Male 0 Female 0

T1P1 Short tail, no preference 500 500 500 500

T1P2 Short tail, preference 0 500 0 500

T2P1 Long tail, no preference 500 1000 490 990

T2P2 Long tail, preference 0 1000 10 1000

5. Set the initial genotype frequencies to those shown below:

In this population, 10% of the females would prefer to mate with a long-tailed
male, although almost the entire population consists of short-tailed males.
Approximately 10% of the males also carry the allele for preference, even
though they do not express it, but almost all the males have short tails. One
lone male has mutated to have a tail that is longer than the others.

To make these initial conditions a little more plausible, we need to allow that
the P2 allele does not confer absolute preference—otherwise the females that car-
ried it up to this generation would not have mated, and the allele would have
been lost. Resetting the mate selection preferences as shown below will give us
females who would strongly prefer long-tailed males but who will settle for
short-tailed males in a pinch.

Clear your previous results from cells O4–V18 and run your macro to see what
happens to genotype frequencies over 15 generations. Do these initial condi-
tions result in runaway sexual selection?

6. Can runaway sexual selection occur when P2 = 0? Set your initial conditions so
that all females and 995 males in the population have the genotype T1P1 and 5
males have the genotype T2P1. Then set the mate selection preferences as shown
below:

2
3
4
5
6
7
8

A B C D E F

Tally Tally

Genotype Male 0 Female 0

T1P1 Short tail, no preference 900 900 900 900

T1P2 Short tail, preference 99 999 100 1000

T2P1 Long tail, no preference 0 999 0 1000

T2P2 Long tail, preference 1 1000 0 1000

PARENTAL GENOTYPE NUMBERS

2
3
4
5
6
7
8

G H I J K L

T1P1 T1P2 T2P1 T2P2

T1P1 0.25 0.25 0.25 0.25

Female T1P2 0.01 0.01 0.49 0.49

genotype T2P1 0.25 0.25 0.25 0.25

T2P2 0.01 0.01 0.49 0.49

Male genotype

MATE SELECTION PREFERENCES

2
3
4
5
6
7
8

G H I J K L

T1P1 T1P2 T2P1 T2P2

T1P1 0.3 0 0.7 0

Female T1P2 0 0 0 0

genotype T2P1 0.3 0 0.7 0

T2P2 0 0 0 0

Male genotype

MATE SELECTION PREFERENCES

Clear your previous results from cells O4–V18 and run your macro to see what
happens to genotype frequencies over 15 generations. Analyze your model out-
puts and explain how this might occur in nature.

7. Genetic drift can also influence changes in allele frequencies over time. To
examine the effects of genetic drift on this model of sexual selection, we are
going to manipulate the population size by running the model using different
initial genotype numbers. For example, instead of having 1000 individuals of
each sex, start with 500 individuals of each. As with our initial conditions, sim-
ply start with equal numbers of each genotype (so in this case, each genotype
number will be 125). Clear your results from your last simulation, set the mate
selection preferences in cells I5–L8 back to their initial values, and run your
model. Record your results and then run the model a few more times, each time
changing the total genotype numbers (but make sure there are equal numbers
of each genotype). What effect does changing the population size have on the
outcome of the model? 

